

1. Report No. FHWA/TX-85 +469-1F		2. Government Accession No.		3. Recipient's Catalog No.	
4. Title and Subtitle Automatic Retrieval of ATR Data				5. Report Date August 1985	
				6. Performing Organization Code	
7. Author(s) Jae Y. Lee and Heon Y. Yeom				8. Performing Organization Report No. Research Report 469-1F	
9. Performing Organization Name and Address Texas Transportation Institute The Texas A&M University System College Station, TX 77843				10. Work Unit No.	
				11. Contract or Grant No. Study No. 2-10-85-469	
12. Sponsoring Agency Name and Address State Department of Highways and Public Transportation Transportation Planning Division P.O. Box 5051 Austin, TX 78763				13. Type of Report and Period Covered Final September 1984 August 1985	
				14. Sponsoring Agency Code	
15. Supplementary Notes Research performed in cooperation with FHWA, U.S. DOT. Research Study Title: Automatic Retrieval of ATR Data					
16. Abstract This report describes and presents the results of a study to develop a microcomputer based software to retrieve data automatically from the Automatic Traffic Recorders, ATR. The computer program presented in this report may be used on many other CP/M based microcomputer systems running MBASIC. The program has been developed in a high level language, MBASIC, for simple maintenance and modification at a later date.					
17. Key Words Automatic Data Retrieval, Microcomputer Telecommunications, CP/M			18. Distribution Statement No Restrictions. This document is available to the public through the National Technical Information Service 5285 Port Road, Springfield, Virginia 22161.		
19. Security Classif. (of this report) Unclassified		20. Security Classif. (of this page) Unclassified		21. No. of Pages 31	22. Price



AUTOMATIC RETRIEVAL OF ATR DATA

by

**Jae Y. Lee
Systems Analyst**

and

**Heon Y. Yeom
Research Assistant**

**Research Report 469-1F
2-10-85-469**

Sponsored by

Texas State Department of Highways and Public Transportation

**In Cooperation with
The U.S. Department of Transportation
Federal Highway Administration**

**Texas Transportation Institute
The Texas A&M University System
College Station, Texas**

August 1985




METRIC CONVERSION FACTORS

Approximate Conversions to Metric Measures

Symbol	When You Know	Multiply by	To Find	Symbol
LENGTH				
in	inches	*2.5	centimeters	cm
ft	feet	30	centimeters	cm
yd	yards	0.9	meters	m
mi	miles	1.6	kilometers	km
AREA				
in ²	square inches	6.5	square centimeters	cm ²
ft ²	square feet	0.09	square meters	m ²
yd ²	square yards	0.8	square meters	m ²
mi ²	square miles	2.6	square kilometers	km ²
	acres	0.4	hectares	ha
MASS (weight)				
oz	ounces	28	grams	g
lb	pounds	0.45	kilograms	kg
	short tons (2000 lb)	0.9	tonnes	t
VOLUME				
tsp	teaspoons	5	milliliters	ml
Tbsp	tablespoons	15	milliliters	ml
fl oz	fluid ounces	30	milliliters	ml
c	cups	0.24	liters	l
pt	pints	0.47	liters	l
qt	quarts	0.95	liters	l
gal	gallons	3.8	liters	l
ft ³	cubic feet	0.03	cubic meters	m ³
yd ³	cubic yards	0.76	cubic meters	m ³
TEMPERATURE (exact)				
°F	Fahrenheit temperature	5/9 (after subtracting 32)	Celsius temperature	°C

Approximate Conversions from Metric Measures

Symbol	When You Know	Multiply by	To Find	Symbol
LENGTH				
mm	millimeters	0.04	inches	in
cm	centimeters	0.4	inches	in
m	meters	3.3	feet	ft
m	meters	1.1	yards	yd
km	kilometers	0.6	miles	mi
AREA				
cm ²	square centimeters	0.16	square inches	in ²
m ²	square meters	1.2	square yards	yd ²
km ²	square kilometers	0.4	square miles	mi ²
ha	hectares (10,000 m ²)	2.5	acres	
MASS (weight)				
g	grams	0.035	ounces	oz
kg	kilograms	2.2	pounds	lb
t	tonnes (1000 kg)	1.1	short tons	
VOLUME				
ml	milliliters	0.03	fluid ounces	fl oz
l	liters	2.1	pints	pt
l	liters	1.06	quarts	qt
l	liters	0.26	gallons	gal
m ³	cubic meters	35	cubic feet	ft ³
m ³	cubic meters	1.3	cubic yards	yd ³
TEMPERATURE (exact)				
°C	Celsius temperature	9/5 (then add 32)	Fahrenheit temperature	°F



* 1 in = 2.54 (exactly). For other exact conversions and more detailed tables, see NBS Misc. Publ. 286, Units of Weights and Measures, Price \$2.25, SD Catalog No. C13.10:286.

ABSTRACT

This report describes and presents the results of a study to develop a microcomputer based software to retrieve data automatically from the Automatic Traffic Recorders, ATR. The computer program presented in this report may be executed on a KAYPRO-2 microcomputer and with only a minor modification, may be used on many other CP/M based microcomputer systems running MBASIC. The program has been developed in a high level language, MBASIC, for simple maintenance and modification at a later date.

KEY WORDS: Automatic Data Retrieval, Microcomputer, Telecommunications, CP/M

SUMMARY

This documentation describes the results and a product of research study to develop a microcomputer based software for automating the data retrieval procedure. The computer program presented in this documentation should be a useful tool to communicate and to retrieve data automatically from the permanent count stations. The retrieved data may be stored on a floppy diskette for permanent storage and data analysis.

IMPLEMENTATION

The automated data retrieval system presented in this report should be useful to transportation engineers who need effective tools for acquiring the critical data from the permanent traffic counter stations. It may be used to retrieve the raw data, interrogate the status of the field equipment, print out the reports, and to store the raw data on a floppy diskette for a permanent storage.

DISCLAIMER

The contents of this report reflect the views of the authors who are responsible for the opinions, findings, and conclusions presented herein. The contents do not necessarily reflect the official views or policies of the Federal Highway Administration or the State Department of Highways and Public Transportation. This report does not constitute a standard, specification, or regulation.

TABLE OF CONTENTS

	<u>Page</u>
INTRODUCTION.	1
SITUATION	2
BACKGROUND.	2
PROTOCOL BLOCK FORMAT Descriptions	3
Control Characters and Their Meanings.	3
APPENDIX A. USER'S GUIDE	4
APPENDIX B. PROGRAM LISTINGS	13



INTRODUCTION

Electronic data collection systems are being used by the State Department of Highways and Public Transportation (SDHPT) in many locations throughout the state of Texas. These permanent traffic counters automatically collect and store data continuously. Data is retrieved from these counters through a telecommunication system whereby each permanent station is polled daily. This automatic data collection and telecommunications system has greatly reduced the manual effort in data collection and has increased the accuracy of the collected data.

However, the SDHPT also needs to communicate with each individual permanent counter station to verify the proper functioning of the equipment and to retrieve those data that were missed during the polling procedure. The primary purpose of this study was to develop a microcomputer based software that will automate this communication and data retrieval procedure with each individual permanent count station.

AUTOMATIC RETRIEVAL OF ATR DATA

SITUATION

Currently, many TELAC permanent counters are installed and are under operation. To retrieve the data from TELAC, one must manually dial the phone number of the particular station and send the commands to ask TELAC to respond. Usually the commands used are somewhat complex and the data format which TELAC sends back is hard to understand. This study developed a program to aid in communication with TELAC without any specific knowledge of the equipment required by the user.

BACKGROUND

SDHPT currently operates over 50 permanent count stations equipped with an automatic vehicle counter network. This permanent vehicle counter equipment is made by both Streeter Richardson and Sarasota Automation, and both use the same communication protocol to communicate with other computers. Hence we refer to them as identical equipment and are referred to as TELAC in this report. These traffic recorders operate in a roadside cabinet to collect and store the vehicle counts continuously. These recorders are polled by the SDHPT's Nova-800 minicomputer and the stored data is transferred and stored onto the SDHPT's main computer. The software developed under this research study now enables the SDHPT personnel to communicate with the TELAC stations through a KAYPRO-2 microcomputer via a smart modem.

The protocol used to communicate with the TELAC is divided into three groups. The COMMAND protocol is used to send commands from the main computer to the TELAC. The RESPONSE protocol is used to acknowledge the receipt of a DATA BLOCK or a COMMAND BLOCK. The DATA protocol is used to receive DATA BLOCKS from either the computer or TELAC. All characters are ASCII, one start bit, 7 data bit, even parity, and one stop bit.

PROTOCOL BLOCK FORMAT Descriptions


Command block	"DLE-CMD CHR-ID data ETX-BCC"
Intermediate block	"STX-CMD CHR-ID data ETB-BCC"
Last block	"STX-CMD CHR-ID data ETB-BCC"
Response block	"ACK-CMD CHR-ID-ETX-BCC"
	"NAK-CMD CHR-ID-ETX-BBC"
	"CAN-CMD CHR-ID-ETX-BBC"
	"ENQ-CMD CHR ID-ETX-BBC"

Control Characters and Their Meanings

- "DLE" Indicates start of a new command.
- "ENQ" Indicates a request to send last response to a data block.
- "STX" Indicates the start of a new data block.
- "ACK" Indicates good command or good data block received.
- "NAK" Indicates bad data block received. Send last data block again.
- "CAN" Indicates an invalid command, invalid data request, an invalid character received or that one of the data parameters received is out of range.
- "ETB" Indicates that an intermediate data block is being sent.
- "ETX" Indicates that the last data block is being sent.
- "CMD CHR" Is the TELAC ID and consists of two characters i.e. "01".
- "BCC" Is a block character that is the module "63" sum of all the characters sent, plus "64". This does not include the "STX", "DLE", "ACK", "NAK", or "CAN" characters, but includes the "ETB" or "ETX" characters. This will be a printable ASCII character.



APPENDIX A



USER'S GUIDE

SET UP PROCEDURE

This documentation describes the usage of a program named TELAC.bas, which helps the user to communicate with the SDHPT's permanent traffic counters using a KAYPRO-2 microcomputer. This program is written in Microsoft BASIC-80 language.

The set up procedure may be summarized as follows. First, the proper equipment required includes a KAYPRO-2 computer, a MT212AD smart modem, a printer, and two disks. Connect the modem through the telephone line and connect the data communication line to the computer's J4 serial data I/O port and the parallel printer output port. Use the parallel printer output port to connect the KAYPRO-2 computer with the printer. You may still use the telephone after connecting with the modem computer.

To execute this program, use the KAYPRO-2 computer connected with the smart modem, a program disk which is labeled "TELAC" and a data disk which is labeled "TALAC.dat". Put the program disk in drive A:, the data disk in drive B:, and turn on the computer. After it is booted and Aϕ>-prompt appears on the screen, type "mbasic t" to start the program.

EXAMPLE: >0 mbasic t

The program will then ask which action it is to undertake by displaying the following menu which will be denoted as the start menu throughout this user's guide.

START MENU

- 1) call station
- 2) manual call
- 3) print the file
- 4) exit to CP/M

Press 1, 2, 3, or 4 for the desired action. Usually, you will want to call the station to retrieve some data. In this case, press (1) and the program will prompt the user to type in the station number to be called. A listing of all stations may be obtained by entering <RETURN>. Option (2) may be used to make a phone call to the station whose number is not in the "phone.dat" file. If this option is selected, the area code and the local telephone number of the station to be called is required. The telephone number should be in the form ###--####. Option (3) is used to make a print out of the data which is already stored on the data disk. The program may be terminated by selecting option (4).

Once a station is selected, it will automatically make a phone call to that particular station. If connected correctly, program execution will continue in order to communicate with the TELAC.

Call the station.

In this mode, a station may be called directly by typing the sequential number or the station list is displayed on the screen. When <RETURN> is entered and the station list appears on the screen, there are several options. By typing the station number, you can call the station. To see the next page of the listing just press <RETURN>. If it is desirable to see the previous page, press "-" and the program will display the previous page. After a station is selected, the program will display the selected station and call the correct telephone number. If correctly connected, the program will display the main menu.

The main menu is displayed as follows:

MAIN MENU

- 0) Terminate telephone call
- 1) Send current TELAC status
- 2) Send the requested dat's data
- 3) Count monitor mode
- 4) Loop monitor mode
- 5) Send current date and time
- 6) Set date and time
- 7) Change program recording interval
- 8) RAM and PROM memory diagnostic
- 9) Terminate current command

Select one of these ten options by simply typing the proper number, then the KAYPRO will do the rest of the work. After completion of the desired communication, exit to the start menu by selecting 0. Be sure that the MODEM is reset after the phone call is terminated.

- 0) Terminate the telephone call

When the communication with a particular station is complete, this option will electronically "hang-up" the phone call. There are some instances in which the user has to reset the modem by pushing the reset button on the modem due to a malfunction of the TELAC. After disconnecting the modem and hanging up the phone, it returns to the starting menu.

- 1) Send current TELAC status

This menu is used to ask the TELAC to send its current status. The status report consists of the number of the days which TELAC can store data in its memory, the number of days currently stored, number of data blocks per day (usually 24), and the number of the data channels (usually 2 in normal circumstances).

2) Send the requested day's data

This option is used to retrieve the data stored in TELAC's internal memory. When called, this routine will prompt for which day's data is requested. TELAC can send any day's data stored in its internal memory. To select a day, input a number which is the difference between today and that particular day. For example, 0 for today, 1 for yesterday, 2 for the day before yesterday, etc. Since TELAC can store up to 80 day's data, the number selected must be in the range between 0 and 80.

After sending the request to TELAC, it waits until all the requested data is transferred. An indication of this transfer process is shown by displaying of a dot whenever it receives a block of data. It takes approximately 10 seconds to receive one day's data.

Once it receives all the data, the program asks if it is required to produce a print out of the data or to simply store the data into the data disk. Enter "y" or "n" as required. Make sure the printer is properly connected and turned on before proceeding. If the option to save the data on the disk, is selected, the program would also prompt for the name of the data file on which to store the data. It is suggested to name the data files according to the date (e.g., "AUG-11-85") to provide for consistency and a good record keeping format. The data being stored in this way may be retrieved by choosing option (3) (list the data file) from the main menu any time a data listing is required. Approximately 200 day's data may be stored on a single diskette.

3) Count monitor loop

This routine displays the current time and the counts of each data channel. Usually there are two channels designated. Channels 1 and 2.

4) Loop monitor mode

When called, this routine asks TELAC to send the information whenever the loop detects a vehicle. It would display three characters at a time. For a detailed explanation of these characters, refer to the TELAC manual. To interrupt this loop detector mode, press "x" and wait. Exiting this mode may not be immediate and may take time.

5) Current date and time

This routine displays the current date and time stored in TELAC. If this date or time is incorrect, you may correct them using option (6) from the main menu.

6) Set date and time

This routine is used to change the date and time setting of the TELAC or to initialize it. The correct date and time should be entered according to a format of MM/DD/YY/HH/MM. Be sure to type in the "/" as directed.

7) Change program reading interval

Usually, TELAC is programmed to read and record the data at 15 minute intervals. Only two other intervals (30 or 60 minutes) are possible. These may be selected by pressing 1, 2, or 3 and not by entering 15, 30, or 60.

8) RAM and PROM diagnostic

This routine asks TELAC to send its RAM and PROM status and display it on the screen. The error codes are of the form ####. An error code of 0000 indicates no error. Refer to Table 1. for a detailed description of the error codes.

Table 1. RAM and PROM error codes

Code	Board	IC or IC's
1101 to 1104	CPU bd.	U-17
1105 to 1108	CPU bd.	U-18
1109 to 1112	CPU bd.	U-19
2101 to 2106	8K RAM bd.	U-12
2107 to 2112	8K RAM bd.	U- 6
2201 to 2206	8K RAM bd.	U-11
2207 to 2212	8K RAM bd.	U- 5
2301 to 2306	8K RAM bd.	U-10
2307 to 2312	8K RAM bd.	U- 4
2401 to 2406	8K RAM bd.	U- 9
2407 to 2412	8K RAM bd.	U- 3
3003	CPU bd.	U-7, 8, 9
3004	CPU bd.	U-4, 5, 6


9) Terminate current command

This is to inform the TELAC to terminate the execution of current command.


MAINTENANCE

The program does not need to be modified unless the telephone protocol is changed such that this program is used at the other places. In that case, you should find the segment of the program which makes the pre-dial to access the TEXAN line or access an "outside" line. It is set up to dial "8" first to call the station if it may be reached via the TEXAN line "9" + area code and local number to call the stations elsewhere. You can just find that segment and change the numbers.

Another part of the program which may require modification is the telephone data file. The data file to store the telephone numbers of the stations is called "PHONE.DAT" and is stored on disk A:. When these changes occur on the list of phone numbers the data file must be updated. The data file was written using the word processor WORDSTAR. Using WORDSTAR, phone numbers may be easily inserted, deleted, or modified. To invoke this process, type WS PHONE.DAT. at the CP/M prompt. Refer to the WORDSTAR manual for details on the correct procedures to modify the data file. The data file should be on disk A. Since the diskette is usually write protected, the write-protection tab may have to be removed to update the program or the data. It is strongly recommended that a back-up of the TELAC program disk be made prior to attempting any changes in case of mistakes made during modification.



APPENDIX B



PROGRAM LISTINGS

```

10 REM "....."
20 REM          DATA definitions
30 REM "....."
40 ETX=3:ENQ=5:ACK=6:BEL=7:DLE=16
50 CR=13:NAK=21:CAN=24:ETB=23:STX=2
60 DIM CMD(20),PH$(200),DAT(1000),B$(34)
70 CMD(3)=ASC("0"):CMD(4)=ASC("1")
80 LLL(1)=49:LLL(2)=53:LLL(3)=51
90 LLL(4)=48:LLL(5)=54:LLL(6)=48
100 BL$=""
110 LI$="-----"
120 CD$="HABCDEILPX"
130 FOR I=0 TO 9
140     CM(I)=ASC(MID$(CD$,I+1,1))
150 NEXT
160 REM
170 REM "....."
180 REM          Set Z-80 SID port
190 REM "....."
200 OUT 0,5
210 OUT 6,4:OUT 6,71:      REM 300 baud,even parity
220 OUT 6,5:OUT 6,170:    REM  RTS enable
230 OUT 6,3:OUT 6,65:    REM  Receive enable
240 REM
250 REM "....."
260 REM          Open data file and read it (phone numbers)
270 REM "....."
280 OPEN "i",#1,"PHONE.DAT"
290 INPUT#1,PH$(1)
300 FOR I=1 TO 250
310     INPUT#1,PH$(I)
320     IF PH$(I)="END" THEN GOTO 350
330 NEXT
340 CLOSE#1
350 MAX=I:MX=CINT(MAX/18)
360 REM "....."
370 GOSUB 2200
380 PRINT:PRINT"          1. Call station"
390 PRINT:PRINT"          2. Manual call"
400 PRINT:PRINT"          3. List the data file"
410 PRINT:PRINT"          4. Exit to CP/M"
420 PRINT:PRINT"          select (1/2/3/4) ? ";
430 GOSUB 5940:CC$=Y$
440 PRINT:PRINT:PRINT
450 IF CC$="3" THEN GOTO 690
460 IF CC$="4" THEN SYSTEM
470 IF CC$<>"2" GOTO 550
480 PRINT:PRINT:PRINT:PRINT"          Type in the telephone number to call."
490 PRINT:INPUT"          Area code= ";AR$
500 IF VAL(AR$)=0 OR VAL(AR$)>999 GOTO 450

```

```

510 PRINT:INPUT"      Local telephone number = ";LT$
520 IF LEN(LT$)<>8 GOTO 510
530 B$=AR$+" "+LT$
540 GOTO 880
550 IF CC$<>"1" GOTO 430
560 PRINT "      TYPE THE STATION NUMBER TO CALL (<return> for listing)":PRINT
570 INPUT "      station number --> ";IN$
580 IF IN$="" THEN GOTO 1320
590 I=VAL(IN$)
600 IF I>0 THEN A$=PH$(I):GOTO 860
610 FOR I=1 TO MAX
620 A$=PH$(I)
630 ST$=LEFT$(A$,5)
640 IF ST$=IN$ THEN GOTO 860
650 NEXT
660 PRINT:PRINT" no such station..... try again"
670 PRINT:PRINT:PRINT:PRINT
680 GOTO 560
690 REM "....."
700 REM      Print out data file
710 REM .....
720 PRINT:PRINT:INPUT"Name of the data file = ";F$
730 PRINT:INPUT " Switch the Printer on and press <return> ";Y$
740 OPEN "i",#2,"b:"+F$
750 FOR N=0 TO 33
760 INPUT#2,F$
770 PRINT F$
780 LPRINT" ";F$
790 NEXT
800 CLOSE#2
810 GOTO 370
820 REM .....
830 REM
840 REM      Dialing routine
850 REM.....
860 PRINT:PRINT:PRINT:PRINT:PRINT" selected station : ";A$
870 B$=MID$(A$,31,12)
880 D$="CPRC ATV1"
890 GOSUB 1640:GOSUB 1760
900 D$="ath"
910 GOSUB 1640
920 GOSUB 1760
930 PRINT:PRINT:PRINT:PRINT:PRINT"      Now dialing ";B$:PRINT:PRINT
940 D$="CPRC ATDT"
950 GOSUB 1640
960 AR$=MID$(B$,1,3)
970 IF AR$="512" THEN D$="9,.,." ELSE D$="9,.,."+AR$+","
980 GOSUB 1640
990 D$=MID$(B$,4,9)
1000 GOSUB 1640

```

```

1010 GOSUB 1760
1020 PRINT:PRINT:PRINT
1030 GOSUB 2040
1040 IF A=10 OR A=13 THEN GOTO 1060
1050 GOTO 1030
1060 R$=""
1070 GOSUB 2040
1080 IF A=10 OR A=13 THEN GOTO 1110
1090 R$=R$+CHR$(A)
1100 GOTO 1070
1110 PRINT" =====> ";R$:PRINT
1120 IF R$="CONNECT" THEN GOTO 1190
1130 IF R$="NO CARRIER" OR R$="ERROR" THEN GOTO 1220
1140 IF R$="NO DIAL TONE" OR R$="PHONE BUSY" THEN GOTO 1220
1150 RR=VAL(R$)
1160 IF RR=0 THEN GOTO 1030
1170 ON RR GOTO 1190,1030,1220,1220,1030,1030,1220,1220
1180 GOTO 1030
1190 GOTO 2370
1200 D$="CFRC ATH"
1210 GOSUB 1640:GOSUB 1760
1220 PRINT:PRINT:PRINT"          TRY AGAIN ? (<return> for yes, others for no)";
1230 GOSUB 5940: IF Y$="" GOTO 930
1240 GOTO 860
1250 GOTO 930
1260 REM
1270 REM .....
1280 REM
1290 PRINT:PRINT:PRINT" #          STATION DIST          LOCATIGN          TELEPHONE"
1300 PRINT"-----"
1310 RETURN
1320 FOR II=1 TO MX
1330   GOSUB 1290
1340   IF II<1 THEN II=1
1350   FOR I=1 TO 18
1360     J=II*18-18+I
1370     IF J<10 THEN PRINT" ";
1380     IF J<100 THEN PRINT" ";
1390     IF J=>MAX THEN PRINT:GOTO 1410
1400     PRINT" ";J;SPC(4);FH$(J)
1410   NEXT
1420   PRINT"-----"
1430   PRINT" <return> to continue, '-' to go back, # to dial"
1440   INPUT;" 'x' to exit      :      select one ---> ";IN$
1450   IF IN$="" THEN GOTO 1520
1460   IF IN$="-" THEN II=II-1 :GOTO 1330
1470   IF IN$="X" OR IN$=":" THEN GOTO 1540
1480   J=VAL(IN$)
1490   IF J<1 OR J>MAX THEN GOTO 1360
1500   A$=FH$(J)

```

```

1510     GOTO 860
1520     IF II=MX THEN II=II-1
1530 NEXT
1540 PRINT:PRINT:PRINT"         want to scan again (y/n) ";
1550 GOSUB 5940
1560 IF Y$="y" OR Y$="Y" GOTO 1290
1570 GOTO 370
1580 REM
1590 REM .....
1600 REM
1610 REM         Send a Dialing command to MODEM
1620 REM
1630 REM.....
1640 FOR J=1 TO LEN(D$)
1650 GOSUB 1960
1660 C$=MID$(D$,J,1)
1670 GOSUB 5790
1680 OUT 4,ASC(C$)
1690 NEXT
1700 RETURN
1710 REM
1720 REM .....
1730 REM
1740 REM         Send CR to MODEM
1750 REM .....
1760 OUT 4,13
1770 RETURN
1780 REM ".....
1790 REM         Send a Character to MODEM
1800 REM .....
1810 REM
1820 FT=0
1830 FOR K=0 TO 6
1840     TEST=2^K
1850     BIT=A AND TEST
1860     IF BIT = 1 THEN FT=FT+1: IF FT=2 THEN FT=0
1870 NEXT
1880 IF FT=1 THEN A=A OR 128
1890 GOSUB 1960
1900 OUT 4,A
1910 RETURN
1920 REM
1930 REM .....
1940 REM         waits MODEM is ready to receive
1950 REM .....
1960 B=INP(6)
1970 B=B AND 4
1980 IF B=0 THEN GOTO 1960
1990 RETURN
2000 REM

```

```

2010 REM *****
2020 REM           Get a character from the MODEM
2030 REM .....
2040 B=INP(6)
2050 B=B AND 1
2060 IF B=0 GOTO 2040
2070 A=INP(4) AND 7*16+15
2080 RETURN
2090 CTC$=INKEY$
2100 IF LEN(CTC$)=0 GOTO 2050
2110 IF ASC(CTC$)=4 GOTO 2130
2120 GOTO 2050
2130 A=INP(4)
2140 A=A AND 7*16+15
2150 RETURN
2160 REM
2170 REM .....
2180 REM           print the start screen
2190 REM .....
2200 PRINT CHR$(26)
2210 PRINT:PRINT:PRINT
2220 PRINT" *****:
2230 PRINT" *:
2240 PRINT" *           Let's talk to TELAC:
2250 PRINT" *:
2260 PRINT" *****:
2270 PRINT:PRINT
2280 PRINT:PRINT:PRINT
2290 RETURN
2300 REM
2310 REM .....
2320 REM
2330 REM
2340 REM           MAIN PROGRAM to communicate with TELAC
2350 REM
2360 REM .....
2370 M$(1)=" SEND CURRENT TELAC STATUS"
2380 M$(2)=" SEND THE REQUESTED DAY's DATA"
2390 M$(3)=" COUNT MONITOR MODE"
2400 M$(4)=" LOOP MONITOR MODE"
2410 M$(5)=" SEND CURRENT DATE AND TIME"
2420 M$(6)=" SET DATE AND TIME"
2430 M$(7)=" CHANGE PROGRAM RECORDING INTERVAL"
2440 M$(8)=" RAM AND FROM MEMORY DIAGNOSTICS"
2450 M$(9)=" TERMINATE CURRENT COMMAND"
2460 M$(0)=" TERMINATE TELEPHONE CALL"
2470 T$="MASTER COMMAND MENU"
2480 GOSUB 5710
2490 FOR I=0 TO 9
2500 PRINT

```

```

2510 PRINT " ";I;". ";M$(I)
2520 NEXT
2530 PRINT:PRINT" -----"
2540 PRINT:PRINT" SELECT ONE (0-9) : ";
2550 I$=INKEY$: IF LEN(I$)=0 THEN GOTO 2550
2560 IF I$<"0" OR I$>"9" THEN PRINT CHR$(7):GOTO 2550
2570 PRINT I$
2580 CC=VAL(I$)
2590 CMD(0)=5
2600 CMD(1)=DLE
2610 CMD(2)=CM(CC)
2620 CMD(5)=ETX
2630 IF CC=2 THEN GOSUB 2940
2640 IF CC=6 THEN GOSUB 3230
2650 IF CC=7 THEN GOSUB 3430
2660 GOSUB 2780
2670 GOSUB 3590
2680 GOTO 2710
2690 FOR I=1 TO P:LPRINT DAT(I),CHR$(DAT(I)):PRINT CHR$(DAT(I));:NEXT
2700 LPRINT:LPRINT
2710 ON CC+1 GOSUB 5800,4290,4490,5080,5460,5270,5460,5460,5500,5460
2720 PRINT:PRINT:PRINT
2730 PRINT:PRINT:PRINT" PRESS ANY KEY TO CONTINUE"
2740 GOSUB 5920
2750 GOTO 2470
2760 REM
2770 REM .....
2780 REM .....
2790 REM
2800 REM Send command to TELAC
2810 REM .....
2820 BB=-CMD(1)
2830 FOR I=1 TO CMD(0)
2840 A=CMD(I)
2850 BB=BB+A
2860 GOSUB 1810
2870 NEXT
2880 IF BB<64 THEN B=B+64
2890 IF BB<128 THEN GOTO 2910
2900 BB=BB-64 : GOTO 2890
2910 A=BB
2920 GOSUB 1810
2930 RETURN
2940 REM
2950 REM .....
2960 REM Subroutine to make "B" command.
2970 REM .....
2980 FOR I=5 TO 8
2990 CMD(I)=48
3000 NEXT

```



```

3010 MB$(0)="      0.  TODAY"
3020 MB$(1)="      1.  YESTERDAY"
3030 MB$(2)="      2.  2 Days before"
3040 MB$(3)="      N.  N Days before (up to 80)"
3050 T$="SEND THE REQUESTED DAT'S DATA"
3060 GOSUB 5710
3070 PRINT:PRINT:PRINT
3080 FOR I=0 TO 3
3090     PRINT  MB$(I)
3100     PRINT
3110     IF I=2 THEN PRINT"      .":PRINT"      .":PRINT
3120 NEXT
3130 PRINT:PRINT
3140 PRINT:PRINT" Select (0,1,2,.....,80) ? ";
3150 INPUT;D$
3160 IF D$="0" GOTO 3200
3170 IF VAL(D$)<=0 OR VAL(D$)>80 GOTO 3140
3180 IF VAL(D$)>10 THEN CMD(7)=ASC(LEFT$(D$,1))
3190 CMD(8)=ASC(RIGHT$(D$,1))
3200 CMD(9)=ETX
3210 CMD(0)=9
3220 RETURN
3230 REM .....
3240 REM
3250 REM          SUBROUTINE TO MAKE "I" command
3260 REM
3270 REM .....
3280 T$="SET DATE AND TIME"
3290 GOSUB 5710
3300 PRINT:PRINT:PRINT
3310 PRINT" TYPE IN DATE AND TIME (MM/DD/YY/HH/MM)"
3320 INPUT"          ";DT$
3330 IF LEN(DT$)<>14 THEN GOTO 3300
3340 J=1
3350 FOR I=5 TO 13 STEP 2
3360 CMD(I)=ASC(MID$(DT$,J,1))
3370 CMD(I+1)=ASC(MID$(DT$,J+1,1))
3380 J=J+3
3390 NEXT
3400 CMD(15)=ETX
3410 CMD(0)=15
3420 RETURN
3430 REM .....
3440 REM          Make "L" command
3450 REM .....
3460 T$="Change Program Recording Interval"
3470 GOSUB 5710
3480 PRINT:PRINT"      1 : 15 minutes"
3490 PRINT:PRINT"      2 : 30 minutes"
3500 PRINT:PRINT"      3 : 60 minutes"

```

```

3510 PRINT:PRINT :PRINT " select new interval (1-3) ";
3520 GOSUB 5940
3530 LL=VAL(Y$)
3540 CMD(5)=LLL(2*LL-1)
3550 CMD(6)=LLL(2*LL)
3560 CMD(7)=ETX
3570 CMD(0)=7
3580 RETURN
3590 REM
3600 REM .....
3610 REM
3620 REM          Subroutine to receive DATA from TELAC
3630 REM
3640 REM .....
3650 PRINT:PRINT" TELAC Responds .";
3660 IF CC=4 GOTO 4060
3670 F=1
3680 GOSUB 2040
3690 PRINT". ";
3700 IF A=STX THEN GOTO 3750
3710 IF A=ACK GOTO 3750
3720 IF A=NAK GOTO 3750
3730 IF A=CAN GOTO 3750
3740 IF A=ENQ GOTO 3750
3750 DAT(F)=A
3760 F=F+1
3770 GOSUB 2040
3780 DAT(F)=A
3790 F=F+1
3800 IF A=ETB THEN PRINT"..";: GOSUB 4230
3810 IF A>ETX THEN GOTO 3770
3820 GOSUB 2040
3830 PRINT:PRINT
3840 GOSUB 4230
3850 RETURN
3860 IF FIRST=ENQ THEN GOTO 2660: REM re-send
3870 IF FIRST=CAN THEN GOTO 4000
3880 IF FIRST=NAK THEN GOTO 2660: REM re-send
3890 IF FIRST=ACK THEN GOTO 3950
3900 IF FIRST=STX THEN GOTO 3950
3910 PRINT:PRINT:PRINT"          Unrecognized response"
3920 PRINT:PRINT:PRINT " Press any key to continue";
3930 GOSUB 5920
3940 GOTO 2350
3950 CMD(0)=5
3960 CMD(1)=ACK
3970 CMD(5)=ETX
3980 GOSUB 2780
3990 RETURN
4000 IF CC=2 THEN PRINT:PRINT:PRINT" Data is not available"
      ELSE PRINT:PRINT:PRINT" Data is out
of range"

```

```

4010 PRINT:PRINT
4020 PRINT:PRINT:PRINT" Press any key to continue";
4030 GOTO 2470
4040 REM
4050 REM "....."
4060 REM          SHOW THE LOOP MONITOR OUTPUT
4070 REM .....
4080 T$="Loop Monitor Mode"
4090 GOSUB 5710
4100 PRINT:PRINT"Press any key to stop monitoring"
4110 PRINT:PRINT
4120 GOSUB 2040
4130 PRINT CHR$(A);
4140 CTC$=INKEY$:IF LEN(CTC$)>0 GOTO 4160
4150 GOTO 4120
4160 CMD(2)=ASC("X")
4170 GOSUB 2820
4180 RETURN
4190 REM
4200 REM "....."
4210 REM          Send ACK signal to TELAC
4220 REM .....
4230 CMD(1)=ACK: CMD(5)=ETX: CMD(0)=5
4240 GOSUB 2820
4250 RETURN
4260 REM "....."
4270 REM          display "A" output
4280 REM .....
4290 T$="CURRENT TELAC STATUS"
4300 GOSUB 5710
4310 PRINT:PRINT:PRINT
4320 AA$(0)="# of DAY's MEMORY           "
4330 AA$(1)="# of DAY's stored           "
4340 AA$(2)="# of DATA blocks per day  "
4350 AA$(3)="# of DATA CHANNELs       "
4360 AA$(4)="Recording Interval        "
4370 FOR K=1 TO 5: IF DAT(K)=STX GOTO 4380:NEXT
4380 K=K+4
4390 FOR I=0 TO 3
4400   AS$(I)=" : "
4410   FOR J=1 TO 4
4420     AS$(I)=AS$(I)+CHR$(DAT(I*4+J+K))
4430   NEXT
4440 NEXT
4450 FOR I=0 TO 3
4460   PRINT:PRINT SPC(5);AA$(I);AS$(I)
4470 NEXT
4480 RETURN
4490 REM
4500 REM "....."

```

```

4510 REM
4520 REM          Show the "B" command output
4530 REM .....
4540 MM=1
4550 PRINT:INPUT" Do you want to make the print out (y/n)?"P$
4560 IF P$="y" OR P$="Y" THEN PRINT:PRINT" Turn the Printer on and";:GOSUB 5970
4570 PRINT:PRINT"          Date          Time          Channel 1  Channel 2 "
4580 PRINT" -----"
4590 IF P$<>"y" AND P$<>"Y" GOTO 4740
4600 B$(0)=BL$
4610 B$(1)="          Station : "
4620 B$(2)=BL$
4630 B$(3)="          "+A$
4640 B$(4)=BL$
4650 B$(5)=BL$
4660 B$(6)=LI$
4670 B$(7)="          Date          Time          Channel 1  Channel 2"
4680 B$(8)=LI$
4690 B$(9)=BL$
4700 IF P$<>"y" AND P$<>"Y" GOTO 4740
4710 FOR I=0 TO 9
4720   LPRINT B$(I)
4730 NEXT
4740 FOR I=10 TO 33
4750 FOR M=MM TO MM+5 : IF DAT(M)=STX GOTO 4760: NEXT M
4760 IF DAT(M+1) = ASC("B") THEN MM=M+3 ELSE MM=MM+1:GOTO 4750
4770   BB$=CHR$(DAT(MM+1))+CHR$(DAT(MM+2))
4780   FOR J=3 TO 5 STEP 2
4790     BB$=BB$+"/"+CHR$(DAT(MM+J))+CHR$(DAT(MM+J+1))
4800   NEXT J
4810   BB$=BB$+"          "
4820   BB$=BB$+CHR$(DAT(MM+7))+CHR$(DAT(MM+8))+": "
4830   BB$=BB$+CHR$(DAT(MM+9))+CHR$(DAT(MM+10))
4840   BB$=BB$+"          "
4850   FOR J=11 TO 14
4860     BB$=BB$+CHR$(DAT(MM+J))
4870   NEXT J
4880   BB$=BB$+"          "
4890   FOR J=15 TO 18
4900     BB$=BB$+CHR$(DAT(MM+J))
4910   NEXT J
4920   B$="          "+BB$
4930   PRINT"          ";BB$
4940   IF P$="y" OR P$="Y" THEN LPRINT "          ";BB$
4950 MM=MM+18
4960 NEXT I
4970 PRINT:PRINT:PRINT" Do you want to save it on the disk (y/n)?"DI$
4980 GOSUB 5940: IF Y$="y" OR Y$="Y" GOTO 5000
4990 RETURN
5000 PRINT:PRINT" Name of the file to save : ";

```

```

5010 INPUT;F$
5020 OPEN "O",#2,"b:"+F$
5030 FOR I=0 TO 33
5040     WRITE #2,B$(I)
5050 NEXT
5060 CLOSE#2
5070 RETURN
5080 REM "....."
5090 REM     show the "C" command output
5100 REM .....
5110 T$="Count monitor mode"
5120 GOSUB 5710
5130 PRINT:PRINT:PRINT
5140 PRINT:PRINT:PRINT"           Current time = ";
5150 PRINT CHR$(DAT(5))+CHR$(DAT(6))+":"+CHR$(DAT(7))+CHR$(DAT(8))
5160 PRINT:PRINT"           direction 1 Count = ";
5170 FOR I=9 TO 12
5180     PRINT CHR$(DAT(I));
5190 NEXT
5200 PRINT
5210 PRINT:PRINT"           direction 2 Count = ";
5220 FOR I=13 TO 16
5230     PRINT CHR$(DAT(I));
5240 NEXT
5250 RETURN
5260 REM
5270 REM "....."
5280 REM     show "E" command output
5290 REM .....
5300 T$="CURRENT DATE AND TIME"
5310 GOSUB 5710
5320 PRINT:PRINT:PRINT
5330 PRINT:PRINT"           Date : ";
5340 FOR I=5 TO 9 STEP 2
5350     PRINT CHR$(DAT(I));CHR$(DAT(I+1));
5360     IF I<>9 THEN PRINT"/";
5370 NEXT
5380 PRINT:PRINT
5390 PRINT:PRINT"           Time : ";
5400 FOR I=11 TO 14
5410     PRINT CHR$(DAT(I));
5420     IF I=12 THEN PRINT":";
5430 NEXT
5440 PRINT:PRINT
5450 RETURN
5460 REM "....."
5470 REM     No output.. Return to MAIN loop
5480 REM .....
5490 GOTO 2700
5500 REM "....."

```

```

5510 REM          show "F" command output
5520 REM .....
5530 T$="RAM and FROM Memory Diagnostics"
5540 GOSUB 5710
5550 ER=0
5560 PRINT:PRINT"          RAM error Code  : ";
5570 FOR I= 5 TO 8
5580     PRINT CHR$(DAT(I));
5590 IF DAT(I)<>48 THEN ER=ER+1
5600 NEXT
5610 PRINT:PRINT:PRINT"          FROM error Code  : ";
5620 FOR I=9 TO 12
5630     PRINT CHR$(DAT(I));
5640 IF DAT(I)<>48 THEN ER=ER+1
5650 NEXT
5660 PRINT:PRINT
5670 IF ER=0 THEN PRINT"          Everything's O.K." ELSE PRINT"          Something's wro
g. Please check !"
5680 PRINT:PRINT
5690 RETURN
5700 END
5710 PRINT CHR$(26)
5720 PRINT:PRINT
5730 T=10
5740 PRINT SPC(T);T$
5750 PRINT" -----"
5760 RETURN
5770 PRINT B:END
5780 GOTO 2370
5790 FOR I=1 TO 100 : NEXT : RETURN
5800 REM
5810 REM "....."
5820 REM          Hang up the phone and bye ....
5830 REM .....
5840 PRINT:PRINT:PRINT"          Are you sure ? (y/n)";
5850 GOSUB 5940
5860 IF Y$="y" OR Y$="Y" THEN GOTO 5880
5870 RETURN
5880 D$="CFRC ATH"
5890 GOSUB 1640
5900 GOSUB 1760
5910 GOTO 370
5920 Y$=INKEY$: IF LEN(Y$)=0 GOTO 5920
5930 RETURN
5940 Y$=INKEY$: IF LEN(Y$)=0 GOTO 5940
5950 PRINT Y$
5960 RETURN
5970 PRINT" hit any key to continue"
5980 GOSUB 5920
5990 RETURN

```